# Maintainer: Sven-Hendrik Haase <svenstaro@gmail.com>

pkgname=python-pynvim
pkgver=0.4.3
pkgrel=3
pkgdesc='Python client for Neovim'
url='https://github.com/neovim/pynvim'
arch=('any')
license=('Apache')
depends=('python' 'python-msgpack' 'python-greenlet')
provides=('python-neovim')
makedepends=('python-setuptools')
replaces=('python-neovim')
source=("$pkgname-$pkgver.tar.gz::https://github.com/neovim/pynvim/archive/${pkgver}.tar.gz")
sha256sums=('e7c9de44b0201ad874a608270b7a9b10fd48bda65f49bada05815d973ca79391')

prepare() {
  sed -i "s/msgpack-python/msgpack/g" pynvim-${pkgver}/setup.py
}

build() {
  cd "${srcdir}/pynvim-${pkgver}"
  python setup.py build
}

package() {
  cd "${srcdir}/pynvim-${pkgver}"
  python setup.py install --optimize=1 --prefix=/usr --root="${pkgdir}" --skip-build
}

# vim:set sw=2 sts=2 et:
